

Postestimation commands

The following postestimation commands are available after `frontier`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of parameters
<code>estat ic</code>	Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICc, and BIC, respectively)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
<code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of parameters
<code>linktest</code>	link test for model specification
<code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of parameters
<code>predict</code>	predictions and their SEs, technical efficiency, etc.
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of parameters
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

predict

Description for predict

`predict` creates a new variable containing predictions such as linear predictions, standard errors, and estimates of technical efficiency.

Menu for predict

Statistics > Postestimation

Syntax for predict

```
predict [type] newvar [if] [in] [ , statistic ]
```

```
predict [type] stub* [if] [in] , scores
```

<i>statistic</i>	Description
------------------	-------------

Main

xb	linear prediction; the default
stdp	standard error of the prediction
u	estimates of minus the natural log of the technical efficiency via $E(u_i \epsilon_i)$
m	estimates of minus the natural log of the technical efficiency via $M(u_i \epsilon_i)$
te	estimates of the technical efficiency via $E\{\exp(-su_i) \epsilon_i\}$
$s = \begin{cases} 1, & \text{for production functions} \\ -1, & \text{for cost functions} \end{cases}$	

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

Options for predict

Main

`xb`, the default, calculates the linear prediction.

`stdp` calculates the standard error of the linear prediction.

`u` produces estimates of minus the natural log of the technical efficiency via $E(u_i | \epsilon_i)$.

`m` produces estimates of minus the natural log of the technical efficiency via $M(u_i | \epsilon_i)$.

`te` produces estimates of the technical efficiency via $E\{\exp(-su_i) | \epsilon_i\}$.

`scores` calculates equation-level score variables.

The first new variable will contain $\partial \ln L / \partial (\mathbf{x}_i \beta)$.

The second new variable will contain $\partial \ln L / \partial (\ln \text{sig}2v)$.

The third new variable will contain $\partial \ln L / \partial (\ln \text{sig}2u)$.

`scores` may not be specified after estimation with option `distribution(tnormal)`.

margins

Description for margins

`margins` estimates margins of response for linear predictions.

Menu for margins

Statistics > Postestimation

Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [options]
```

<i>statistic</i>	Description
<code>xb</code>	linear prediction; the default
<code>stdp</code>	not allowed with margins
<code>u</code>	not allowed with margins
<code>m</code>	not allowed with margins
<code>te</code>	not allowed with margins

Statistics not allowed with margins are functions of stochastic quantities other than $e(b)$.

For the full syntax, see [\[R\] margins](#).

Remarks and examples

► Example 1


In [example 2](#) of [\[R\] frontier](#), we modeled heteroskedasticity by specifying the `vhet()` option. We would like to compare the predicted efficiency in that case with respect to a model specification without accounting for the presence of heteroskedasticity in the error term. [Kumbhakar and Lovell \(2000, 117\)](#) show that failing to account for heteroskedasticity associated with firm size may lead to bias in the estimation of the technical efficiency. By incorrectly assuming homoskedasticity, the estimates for relatively small firms would be biased upward, while the estimates for relatively large firms would be biased downward. Let's refit the model and use the `te` option of `predict`:

```
. use https://www.stata-press.com/data/r19/frontier1
. frontier lnoutput lnlabor lncapital, vhet(size)
  (output omitted)
. predict te_vhet, te
```

Next, we fit the model assuming homoskedasticity and then again predict the technical efficiency with the `te` option of `predict`:

```
. frontier lnoutput lnlabor lncapital
  (output omitted)
. predict te, te
```

The graph below shows the estimates for technical efficiency for the smaller and larger firms. The technical efficiency tends to be smaller for smaller firms when the model specification accounts for the presence of heteroskedasticity, whereas the predictions for the technical efficiency tends to be smaller for larger firms assuming homoskedasticity. These results agree with the theoretical statement in [Kumbhakar and Lovell \(2000\)](#) because the firm size was actually relevant to model heteroskedasticity in the idiosyncratic component of the error term.



► Example 2

We also test in [example 2](#) of [\[R\] frontier](#) whether the firms use constant returns to scale. We can use `lincom` as an alternative to perform an equivalent test based on the normal distribution.

```
. use https://www.stata-press.com/data/r19/frontier1, clear
. frontier lnoutput lnlabor lncapital, vhet(size)
  (output omitted)
. lincom _b[lnlabor] + _b[lncapital]-1
  (1)  [lnoutput]lnlabor + [lnoutput]lncapital = 1
```

lnoutput	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
(1)	.1022278	.5888511	0.17	0.862	-1.051899	1.256355

The p -value is exactly the same as the one we obtained with the test command in [example 2](#) of [\[R\] frontier](#). However, notice that by using `lincom`, we obtained an estimate of the deviation from the constant returns-to-scale assumption, which is not significantly different from zero in this case.



Reference

Kumbhakar, S. C., and C. A. K. Lovell. 2000. *Stochastic Frontier Analysis*. Cambridge: Cambridge University Press.
<https://doi.org/10.1017/CBO9781139174411>.

Also see

[\[R\] frontier](#) — Stochastic frontier models

[\[U\] 20 Estimation and postestimation commands](#)

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).